

**Gramática do Analisador Sintático
Sem Recursividade à Esquerda
e Com Fatoração à Esquerda**

PROGRAMA ->

program id ;
DECLARACOES_VARIAVEIS
DECLARACOES_DE_SUBPROGRAMAS
COMANDO_COMPOSTO
.

DECLARACOES_VARIAVEIS ->

var LISTA_DECLARACOES_VARIAVEIS
| epsilon

LISTA_DECLARACOES_VARIAVEIS ->

LISTA_DE_IDENTIFICADORES : TIPO ; LISTA_DECLARACOES_VARIAVEIS2

LISTA_DECLARACOES_VARIAVEIS2 ->

LISTA_DE_IDENTIFICADORES : TIPO ; LISTA_DECLARACOES_VARIAVEIS2
| epsilon

LISTA_DE_IDENTIFICADORES ->

id LISTA_DE_IDENTIFICADORES2

LISTA_DE_IDENTIFICADORES2 ->

, **id** LISTA_DE_IDENTIFICADORES2
| epsilon

TIPO ->

integer | real | boolean

DECLARACOES_DE_SUBPROGRAMAS ->

DECLARACOES_DE_SUBPROGRAMAS2

DECLARACOES_DE_SUBPROGRAMAS2 ->

DECLARACAO_DE_SUBPROGRAMA ; DECLARACOES_DE_SUBPROGRAMAS2
| epsilon

DECLARACAO_DE_SUBPROGRAMA ->

procedure id ARGUMENTOS ;
DECLARACOES_VARIAVEIS
DECLARACOES_DE_SUBPROGRAMAS
COMANDO_COMPOSTO

ARGUMENTOS ->

(LISTA_DE_PARAMETROS)
| epsilon

LISTA_DE_PARAMETROS ->

LISTA_DE_IDENTIFICADORES : TIPO LISTA_DE_PARAMETROS2

LISTA_DE_PARAMETROS2 ->

; LISTA_DE_IDENTIFICADORES : TIPO LISTA_DE_PARAMETROS2

| epsilon

COMANDO_COMPOSTO ->

begin COMANDOS_OPCIONAIS **end**

COMANDOS_OPCIONAIS ->

LISTA_DE_COMANDOS

| epsilon

LISTA_DE_COMANDOS ->

COMANDO LISTA_DE_COMANDOS2

LISTA_DE_COMANDOS2 ->

; COMANDO LISTA_DE_COMANDOS2

| epsilon

COMANDO ->

id COMANDO_OPT

| COMANDO_COMPOSTO

| **if** EXPRESSAO **then** COMANDO PARTE_ELSE

| **while** EXPRESSAO **do** COMANDO

COMANDO_OPT ->

:= EXPRESSAO

| (LISTA_DE_EXPRESSOES)

| epsilon

PARTE_ELSE ->

else COMANDO

| epsilon

VARIÁVEL saiu devido a fatoração à esquerda de COMANDO

ATIVACAO_DE_PROCEDIMENTO saiu devido a fatoração à esquerda de COMANDO

LISTA_DE_EXPRESSOES ->

EXPRESSAO LISTA_DE_EXPRESSOES2

LISTA_DE_EXPRESSOES2 ->
 , EXPRESSAO LISTA_DE_EXPRESSOES2
 | epsilon

EXPRESSAO ->
 EXPRESSAO_SIMPLES EXPRESSAO_OPT

EXPRESSAO_OPT ->
 OP_RELACIONAL EXPRESSAO_SIMPLES
 | epsilon

EXPRESSAO_SIMPLES ->
 TERMO EXPRESSAO_SIMPLES2
 | SINAL TERMO EXPRESSAO_SIMPLES2

EXPRESSAO_SIMPLES2 ->
 OP_ADITIVO TERMO EXPRESSAO_SIMPLES2
 | epsilon

TERMO ->
 FATOR TERMO2

TERMO2 ->
 OP_MULTIPLICATIVO FATOR TERMO2
 | epsilon

FATOR ->
 id FATOR_OPT
 | **num_int**
 | **num_real**
 | **true**
 | (EXPRESSAO)
 | **not** FATOR

FATOR_OPT ->
 (LISTA_DE_EXPRESSOES)
 | epsilon

SINAL ->
 + | -

OP_RELACIONAL ->
 = | < | > | <= | >= | <>

OP_ADITIVO ->
 + | - | **or**

OP_MULTIPLICATIVO ->

* | / | **and**